BIOS 511 Lab 1
The Output Delivery System

Please read the following instructions carefully before beginning this lab.

The SAS code provided as a part of this lab includes Windows directory paths that will not exist on your computer. You will need to update the directory paths in order for the SAS code to work.

For this lab, please create a subfolder in the directory where your lab program is stored. You will write all SAS output to this subfolder using the commands taught during the lab. For example, if you are storing your SAS files in "C:\BIOS-511\LAB-01" then create the folder "C:\BIOS-511\LAB-01\OUTPUT" for the SAS output.

For items so marked, include the SAS code in the complete SAS program that you will upload to Sakai to document completion of the lab. After you complete the entire SAS program, save the program (save often during development) and exit your SAS session. Then, reopen SAS and run the entire program from start to finish, verify the SAS log has no unforeseen ERRORS or WARNINGS, and save the SAS log so that you can upload it to Sakai. The program and log should be named lab-01-PID.sas and lab-01-PID.log where PID is your student PID number. You will turn in your SAS program, your SAS program log, and all output files from the lab. Failure to turn in all these files will results in a reduced lab grade.

Your completed SAS program will be a long program with essentially unrelated sections of code. I recommend you comment the code for each section so you can find needed sections later (i.e., to study or as a reference for future assignments). Feel free to separate out the sections of SAS code into separate programs after lab. Constructing a "code library" of example SAS code is very useful, if well documented.

This lab is comprised of a series of enumerated sections. Many of the sections only require you to read and digest information. Others require you to write some SAS code and run it to examine what the code does. For sections that start with "**Include this code in your lab program**" make sure to include the referenced SAS code in the program that you submit at the end of lab. The presence of that code in your lab program is how we will determine if you completed the lab successfully.

BIOS 511 Lab 1
The Output Delivery System

SAS's Output Delivery System

The output delivery system (ODS) provides more visually appealing procedure output than traditional SAS listing output. In fact, SAS Studio (SS) and SAS University Edition (SUE) do not even produce traditional SAS listing output (because it is so antiquated).

Consider the following simple SAS program:

```
proc print data = sashelp.class; run;
```

Prior to SAS version 9.4, when this program is run in SAS Display Manager (SDM) the output would look as follows (by default):

Obs	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69.0	112.5
2	Alice	F	13	56.5	84.0
3	Barbara	F	13	65.3	98.0
4	Carol	F	14	62.8	102.5
5	Henry	M	14	63.5	102.5
6	James	M	12	57.3	83.0
7	Jane	F	12	59.8	84.5
8	Janet	F	15	62.5	112.5
9	Jeffrey	M	13	62.5	84.0
10	John	M	12	59.0	99.5
11	Joyce	F	11	51.3	50.5
12	Judy	F	14	64.3	90.0
13	Louise	F	12	56.3	77.0
14	Mary	F	15	66.5	112.0
15	Philip	M	16	72.0	150.0
16	Robert	M	12	64.8	128.0
17	Ronald	M	15	67.0	133.0
18	Thomas	M	11	57.5	85.0
19	William	M	15	66.5	112.0

Not very pretty! This type of output format was designed for old-style printers. The modern SAS ODS allows programs to direct output to one of several destinations: the listing destination (only supported by the SDM), the PDF destination, the RTF destination, the HTML destination, and to a SAS data set. In addition, the SAS ODS allows programmers to restrict procedure output to just the portions they are really interested in.

Learning Objectives: In this lab, you will learn how to use the SAS ODS for two purposes:

(1) To produce SAS output in various formats: HTML, PDF, output data sets, etc.

(2) To control the output that a given SAS procedure produces.

Note that SS/SUE users have the ability to save output from the results tab as a PDF, HTML, or RTF file. The traditional SDM produces HTML output by default unless instructed otherwise. We will focus on using SAS ODS commands to create PDF files, HTML files, RTF files, and data sets programmatically rather than using the point-click interface provided by SS/SUE. Using SAS ODS commands is critical when working in an environment where there are multiple programmers on the team or when SAS programs need to be executed frequently using different (perhaps updated) data sets as input.

BIOS 511 Lab 1
The Output Delivery System

Output Objects

1. First, you need to know that SAS procedures produce output in structures called *output objects*.

    a. In the SDM, these output objects are listed in the results window (on the left side of the SAS windowing environment) after a procedure step runs.

    b. Unfortunately SS/SUE does not produce such a list, but you can programmatically display the list of objects as described below.

2. **Include this code in your lab program.** Type in and run the code below (do not copy and paste, you need the practice). When the PROC PRINT step is executed, it produces one output object. When this simple PROC UNIVARIATE step is executed, it produces five output objects.

```
proc print data=orion.employee_payroll(obs=10);
        title 'First 10 Observations of Employee Payroll Data Set';
run;

proc univariate data=orion.employee_payroll;
        var salary;
        title 'Descriptive Statistics for Salary';
run;
```

3. *Only applicable if you are using the SDM.* With this SAS modality, you can explore what output objects are produced by viewing the *Results Window*.

Figure 1: Image of the Results Window



Notice that there is one entry in the results tree for each PROC step that was executed. If you expand the tree as much as possible, the lowest-level items (one under PROC PRINT and five under PROC UNIVARIATE) are the output objects.

Double-click on one of these output object entries, such as **Tests for Location** which is nested under **Salary** which is nested under **Univariate: Descriptive Statistics for Item Salary**. Notice that the Results Viewer scrolls to display the output object that you selected. The Results Viewer is showing you HTML output as by default the SDM creates a temporary HTML file that is deleted when the SAS session closes.

Now go back to the Results Window. Right-click on one of the output object entries, and select Properties at the bottom of the action list. The Properties window for an output object tells you its name (e.g., TestsForLocation) and some other potentially useful information.

4. **Include this code in your lab program.** One can programmatically use the `ODS TRACE ON;` and `ODS TRACE OFF;` statements around PROC step code to have output object information printed out to the SAS log.

   This code works in both the SDM and SS/SUE. Try the code from above with `ODS TRACE ON;` and `ODS TRACE OFF;` now surrounding it:

```
ods trace on;
proc print data=orion.employee_payroll(obs=10);
        title 'First 10 Observations of Employee Payroll Data Set';
run;

proc univariate data=orion.employee_payroll;
        var salary;
        title 'Descriptive Statistics for Salary';
run;
ods trace off;
```

   Notice that information on PROC PRINT's one output object and PROC UNIVARIATE's five output objects is written to the SAS Log. One generally uses `ODS TRACE ON;` and `ODS TRACE OFF;` to get information about ODS output objects produced by a procedure but those statements are rarely included in a "final program" (except for this after this lab).

   The default ODS output objects produced by PROC UNIVARIATE are named Moments, BasicMeasures, TestsForLocation, Quantiles, and ExtremeObs.

   Every ODS output object has a unique name (e.g., Quantiles) and output objects can be individually selected for inclusion in a file being created (e.g., a PDF file) or individually excluded from a file being created using their unique name.

BIOS 511 Lab 1
The Output Delivery System

ODS Destinations

5. ODS enables you to instruct SAS that you want to ***deliver*** particular output objects to particular ***destinations***. For example, you might want to produce a two-way frequency table for presentation on the web (i.e., as HTML output), you might want to produce a table of descriptive statistics to include in a MS Word document (i.e., an RTF document), and/or you might want to mail either of these to a colleague.

6. To specify the form in which you want your output delivered by SAS, you use *ODS destinations*. Some of the destinations currently available, along with the ODS statement used to request each one, are listed in the following table:

Destination	Purpose	ODS Statement
HTML files	For display on the web	ODS HTML … ;
PDF files	For printing from the web	ODS PDF … ;
RTF files	To use in word processing	ODS RTF … ;
SAS Output window	SAS listing – not in SS/SUE	ODS LISTING … ;
Output data sets	Produce an output dataset	ODS OUTPUT … ;

These are the destinations you will work with in today's lab. *Note for BIOS students, there is also a Latex destination.*

7. When you run a SAS procedure, it sends its output objects to whatever ODS destinations are currently open (any or all of them can be open at one time).

For the SDM, by default, in SAS 9.4 the HTML destination is open but the file created is temporary and will be deleted upon closing your SAS session. If you want to create a permanent HTML file, you must instruct SAS to do so. If you want some other destination to be used (i.e., PDF), you have to open it before submitting your PROC step(s). To see the output, you need to close the destination after the PROC step(s).

As noted above, the HTML destination is active when you start SAS using the SDM. That means that all output objects will be delivered to the HTML destination unless you instruct SAS to change that behavior. Even if you open the PDF destination, SAS will still deliver objects to the HTML destination (in addition to the PDF destination) unless you instruct SAS to do otherwise.

If you do not want to create an HTML file (or want to stop delivering output to an HTML file being created), you need to submit the statement `ODS HTML CLOSE;` The HTML destination is then closed from that point forward until you instruct SAS to open it back up. This line of code is commonly placed at the top of SAS programs written using the SDM.

For SS/SUE, by default procedure output is written to the Results tab which is also a temporary ODS output storage facility. When you close SS/SUE, the Results tab output is deleted. In fact, any time you submit code in SS/SUE the Results tab is refreshed to only show newly produced output. The output in the Results tab is displayed as HTML but can be downloaded as HTML, PDF, or RTF. You cannot suppress output from being written to the Results tab even if you are separately directing SAS to create permanent files with ODS statements. One can think of the Results tab as being available to provide a quick view of the most recently created output.

After the SAS code from step 4, SDM users please add the `ODS HTML CLOSE;` statement to your SAS program so that there will be no active ODS destinations (from this point forward) in your SAS program. At this point, your SAS program (excluding the header section and preliminary options) should look as follows:

```
/*****************************************************************
                        SAS Code for Section # 2
 *****************************************************************/
.
.
.
/*****************************************************************
                        SAS Code for Section # 4
 *****************************************************************/
.
.
.
ODS HTML CLOSE;
```

USE THIS SECTION COMMENTING CONVENTION FOR THE REMAINDER OF THE LAB.

For SDM users, before going on (and after submitting the code `ODS HTML CLOSE;`), once again submit either the PROC PRINT step or the PROC UNIVARIATE step from item 2 or 4 (highlight just the PROC step you want to run) and observe the message in your SAS log.

You should see the message: `WARNING: No output destinations active`. One will receive such a message any time they run a procedure that produces ODS output when there are no open destinations.

Since SS/SUE *always* writes to the Results Tab, this WARNING is not generally encountered with those SAS modalities.

8. Typically one uses ODS destination statements in pairs, as follows (NOTE: do not try to submit this code – it is incomplete):

```
ods pdf file='path to PDF file you want to create';
/* <procedure steps go here> */
ods pdf close;
```

The first ODS PDF statement instructs SAS that you want subsequent procedure steps to write their output into a PDF file. The second ODS PDF statement closes the PDF destination (i.e., instructs SAS that no more output will be written to this PDF file so that the actual PDF file can be formally created).

Note that this code does not affect any other ODS destination that may be active. That is to say that one can create PDF, RTF, and HTML files all at the same time but only one file of each type may be created at a time.

9. **Include this code in your lab program.** Type in the example below (do not copy and paste, you need the practice, trust me). Before you run the code, make sure you have created the folder specified as the location for the PDF file. SAS will not create a folder that does not exist so that it can create the PDF file within it. SAS simply will not create the PDF file.

```
ods pdf file = 'Path to Your "output" folder/freq9.pdf';
proc freq data=orion.customer;
        tables country gender country*gender;
        title 'Frequency Distributions and Cross-tabulations';
run;
ods pdf close;
```

After running the code, be sure to check the log for errors and correct any programs before moving on. Be sure to close the PDF file when you are finished viewing it. In general if a PDF file is open for viewing (e.g., open in Adobe Acrobat), and you wish to overwrite the contents (perhaps to add more output), SAS cannot overwrite a file unless it is closed.

Adobe Acrobat can be particularly annoying because even when you "close" the program, Windows will often keep the program running in the background on your computer and even keep the PDF file open unbeknownst to you! So, please make sure Adobe Acrobat is actually closed if you use it to view PDF files. What a pain!

If you using the SDM, the PDF file you created will probably pop open automatically in the results viewer window (note: sometimes you will have to close the results viewer before you are able to regenerate results as this effectively keeps the PDF file open).

If you are using SS/SUE, you will have to go into the folder on your computer where the file was created and open the file outside of SAS. The PDF file will not open automatically within SS/SUE. You must still remember to close the file if you wish to recreate it or modify it.

10. **Include this code in your lab program.** Now produce the same output in RTF (rich text format) form, ready for your favorite word processing software. All you need to do is rewrite the same program, but replace "pdf" with "rtf" in the three places where it occurs. Note that instead of the ".rtf" extension, you can specify a ".doc" file extension (but not ".docx"), which corresponds to the extension for an older version of a Microsoft Word document.

```
ods rtf file = 'Path to Your "output" folder/freq10.rtf';
proc freq data=orion.customer;
        tables country gender country*gender;
        title 'Frequency Distributions and Cross-tabulations';
run;
ods rtf close;
```

After running the code, be sure to check the log for errors and correct any programs before moving on. Microsoft Word will open and display your tables, ready for any text you might want to add or modifications you might want to make manually.

Note: To view RTF files correctly on a MAC machine you must open in word processing software that can correctly read files in RTF format. Some MAC word processing programs will incorrectly render the RTF making titles and footnotes not appear correctly.

In your RTF file, did you notice that the output title appeared in the header area rather than as part of the body text? For more flexibility, you'll often want titles and footnotes to be with the body text rather than in the header and footer (especially if you want to copy the title, table, and footnotes to paste into another document such as a manuscript). To accomplish this, use the BODYTITLE option in your opening ODS RTF statement, as follows:

ods rtf file="…" bodytitle;

Repeat the code above with this option and name the file "freq10_bodytitle.rtf".

Does the title "The FREQ Procedure" appear near the top of the output you just produced? If so, you can turn it off in future runs of the program by submitting the global statement:

```
ODS NOPTITLE;
```

To return to the default behavior, you can submit the global statement:

```
ODS PTITLE;
```

Incorporate this feature into your final program by adding the `ODS NOPTITLE;` statement just below the program header. You will seldom want the procedure title in the output (I'm still waiting for such an instance to occur in my life).

11. **Include this code in your lab program.** Creating an HTML file is as simple as making a PDF or RTF file.

```
ods html file = 'Path to Your "output" folder/freq11.html';
proc freq data=orion.customer;
        tables country gender country*gender;
        title 'Frequency Distributions and Cross-tabulations';
run;
ods html close;
```

Again, the only change from the earlier program is substituting "html" for "pdf" in three locations. If you are using the SDM, the produced HTML file will pop open in the results viewer. If using SS/SUE, you can open it outside of SS/SUE using any web browser.

Limiting the Output Produced by a Procedure

12. Sometimes you are only interested in a few of the output objects produced by a SAS procedure. Some SAS procedures, such as PROC UNIVARIATE, yield pretty lengthy output! ODS gives you a way to cut out undesired output.

Use the ODS SELECT statement to specify only the output objects you want.

Use the ODS EXCLUDE statement to exclude only the output objects you don't want.

The syntax of the ODS SELECT/EXCLUDE statements are as follows:

`ODS SELECT` <output object names>;

`ODS EXCLUDE` <output object names>;

In order to find the name of the desired ODS output object, use one of the two methods discussed above (using ODS TRACE ON/OFF is my preferred approach); or simply look in the SAS documentation. For example, google "SAS 9.4 PROC FREQ ODS Table Names". Once you know the names of the ODS output objects you want to include in a file, you can use those names in the ODS SELECT or ODS EXCLUDE statement.

13. **Include this code in your lab** program (do not copy and paste, you need the practice, trust me, I mean it). Compare use of the ODS SELECT and EXCLUDE statements.

```
ods pdf file = 'Path to Your "output" folder/select_exclude13.pdf';

title "Moments and Extreme Observations";
ods select moments extremeobs;
proc univariate data=orion.employee_payroll;
        var salary;
run;

title "Basic Measures, Tests for Location, and Quantiles";
ods exclude moments extremeobs;
proc univariate data=orion.employee_payroll;
        var salary;
run;

title "All Univariate Procedure Output";
proc univariate data=orion.employee_payroll;
        var salary;
run;

ods pdf close;
```

Note the procedure output that is included from the third PROC UNIVARIATE step. What does this imply about ODS SELECT and EXCLUDE statements?

One can also use ALL and NONE instead of specific object names. The default select list is ALL and the default exclude list is NONE. When `ODS SELECT NONE;` is submitted or `ODS EXCLUDE ALL;` is submitted, no output is directed into the file being created. This is useful if all one wants to do is produce a data set of the output for further processing.

Creating Output Data Sets Using ODS

14. You can also use ODS to obtain a data set containing any procedure output. Many procedures provide options or statements for requesting output data sets, but these are limited to certain results. With ODS, you can get <u>any</u> procedure output in the form of a SAS data set.

    To obtain an output data set, you use the ODS OUTPUT statement. That is to say, the OUTPUT destination is used to deliver ODS output objects as SAS data sets for further processing. To use ODS OUTPUT, you need to reference the output objects by name.

    The format of the SAS dataset produced does not always resemble the ODS object you see in printed output.

15. **Include this code in your lab program.** Let's say that you want to deliver the "Quantiles" output object from an analysis of the variable "salary" to a temporary data set named "salary_quant" (remember that one can get the name of the object from looking at the Results Window, by using ODS TRACE ON/OFF; or by googling "SAS 9.4 PROC UNIVARIATE ODS Tables").

```
ods pdf file='Path to Your "output" folder/ods_output_15.pdf';

ods pdf select quantiles;
ods output quantiles = salary quant;
proc univariate data=orion.employee_payroll;
    var salary;
run;

title "Quantiles for Salary Variable";
proc print data = salary_quant; run;

ods pdf close;
```

In this toy example there is no real reason to deliver the "Quantiles" ODS object to a SAS dataset. However, this general strategy will prove useful over the semester when you need to take results from one procedure and post-process them before display or to use as input into another procedure.

Find the WORK library in the explorer window (SDM) or the Libraries menu (SS/SUE), and double-click on the "salary_quant" data set to see the temporary data set that was just created.

Note that many procedures (including the UNIVARIATE Procedure) have mechanisms to delivery specific analysis results to SAS datasets separately from the use of ODS OUTPUT statements. We will learn more about this soon. They key point here is that *any* procedure output can be delivered into a SAS data set using ODS OUTPUT making this a powerful tool.

ODS Styles

16. ODS style templates define the colors, fonts, and font sizes used to create ODS output. SAS provides a default style for each ODS destination. For example, you might have noticed that the HTML files you've created with ODS have used a combination of blue and gray backgrounds. That is the default style for HTML files when using the SDM.

    For the SDM, if you want to see the additional styles that are available, make the following selections from the SAS menu bar:

    Tools → Options → Preferences… → Results tab

    You will see a pull-down list for selecting a default HTML style. Click on the pull-down list to see the available styles; scroll to see the entire list.

    For SS/SUE, you can do the following:

Preferences… → Results tab

You will see a pull-down list for selecting a default style for HTML, PDF, and RTF files. The pull-down lists contain all the styles that are available for SS/SUE.

Both SDM and SS/SUE can programmatically produce a list of available styles using the following PROC step:

```
proc template;
   list styles;
run;
```

No style previewer application is available but you will find them online if you google "SAS 9.4 ODS Styles Gallery".

17. **Include this code in your lab program.** Try out at least two different styles but (in general) only submit assignments using styles that do not require a large amount of dark ink when printed (i.e., do not select a style that has a dark background). Substitute the name of each chosen style in the following code and include your favorite style in the final program:

```
ods pdf file = 'Path to Your "output" folder/fav_style17.pdf' style=journal;
proc freq data=orion.customer;
        tables country gender country*gender;
        title 'Frequency Distributions and Crosstabulations';
run;
ods pdf close;
```

For RTF and PDF files, which are typically printed on non-color printers, styles such as Minimal, Printer, and sansPrinter work well. The Journal style is especially nice for a professional look.

Users can create their own style templates from scratch or by modifying an existing template using The TEMPLATE Procedure. Such topics are beyond the scope of this course. See http://support.sas.com/rnd/base/ods/scratch/styles-tips.pdf for more details.

Page Composition with ODS: The STARTPAGE Option

18. **Include this code in your lab program.** You can use combinations of the options described above, along with a new one, STARTPAGE, to construct nice pages of combined procedure output. The STARTPAGE option controls when SAS inserts a page break (and therefore goes to a new page).

By default, `STARTPAGE=YES`, and SAS inserts a new page at the beginning of each procedure's output. In contrast, `STARTPAGE=NO` is set so that output from multiple procedures goes onto one page. The `STARTPAGE=NO` option instructs SAS to only start a new page if the current page is filled or when you specify `STARTPAGE=NOW`. So if you have set `STARTPAGE=NO` and want to later force SAS to start a new page, you can submit the following statement at the appropriate location:

```
ODS <Destination> STARTPAGE=NOW;
```

Note that STARTPAGE can be used with the PDF or RTF destination only.

Run the following code first without the `STARTPAGE=NO` option set and then with the `STARTPAGE=NO` option set. Lastly, add `ODS PDF STARTPAGE=NOW;` between the PROCs and compare the result to the first case.

```
ods pdf file = 'Path to Your "output" folder/start_page18.pdf'
        style=journal
        startpage=no;

title1 'Descriptive Statistics for Price Variable';
proc means data=orion.employee_payroll;
    class employee_gender;
    var salary;
run;

ods select extremeobs;
proc univariate data=orion.employee_payroll;
    var salary;
run;

ods pdf close;
```

19. **No assistance will be provided for this exercise. Include this code in your lab program.** For this task, you must write a SAS program that uses The CONTENTS Procedure to produce a list of variable names and attributes for three datasets in the orion library: employee_payroll, employee_addresses, and employee_donations. You will direct the output into a PDF file named "contents19.pdf".

    The only output included in the PDF file should be the three ODS objects that list the variable names and attributes (an example is shown below for one dataset).

Alphabetic List of Variables and Attributes					
#	Variable	Type	Len	Format	Label
1	Employee_ID	Num	8	12.	Employee ID
7	Paid_By	Char	17		
2	Qtr1	Num	8		
3	Qtr2	Num	8		
4	Qtr3	Num	8		
5	Qtr4	Num	8		
6	Recipients	Char	65		

    Use the "minimal" ODS style template for the PDF file that you create and ensure that all three ODS objects are included *on the same page* (a one page PDF).