Chapter 9 Logistic Regression II: Polytomous Response

9.1 Introduction

- Chapter 8 discussed logistic regression analysis with a dichotomous response variable; logistic regression is also applicable to multi-level responses

- Response may be ordinal:

  → no pain, slight pain, substantial pain

  → can model *cumulative logits*

  → proportional odds model

  → use PROC LOGISTIC

  or nominal:

  → Democrats, Republicans, Independents

  → can model *generalized logits*

  → use PROC LOGISTIC (or CATMOD)

9.2 Ordinal Response: Proportional Odds Model

9.2.1 Methodology

		Improvement			
Gender	Treatment	Marked	Some	None	Total
Female	Active	16	5	6	27
Female	Placebo	6	7	19	32
Male	Active	5	2	7	14
Male	Placebo	1	0	10	11

- Consider the quantities:

$$\theta_{hi1} = \pi_{hi1}, \qquad \theta_{hi2} = \pi_{hi1} + \pi_{hi2},$$

π_{hi1} denotes probability of marked improvement
π_{hi2} denotes probability of some improvement
π_{hi3} denotes probability of no improvement

- $\{\theta_{hij}\}$ represent cumulative probabilities

$\theta_{hi1} = Pr\{\text{marked improvement}\}$
$\theta_{hi2} = Pr\{\text{marked or some improvement}\}$

$h = 1$ for females, $h = 2$ for males
$i = 1$ for active treatment, $i = 1$ for placebo

- For 3 response levels, compute 2 cumulative logits:

$$\text{logit}(\theta_{hi1}) = \log\left[\frac{\pi_{hi1}}{\pi_{hi2}+\pi_{hi3}}\right]$$

(log odds of marked improvement to none or some)

$$\text{logit}(\theta_{hi2}) = \log\left[\frac{\pi_{hi1}+\pi_{hi2}}{\pi_{hi3}}\right]$$

(log odds of marked or some improvement to none)

- Assume data arise from a stratified simple random sample

- Can write model that applies to both logits simultaneously, allowing different intercept parameters (α_k) and regression parameters (β_k) for each logit:

$$\text{logit}(\theta_{hik}) = \alpha_k + \boldsymbol{x}'_{hi}\boldsymbol{\beta}_k$$

- Proportional odds assumption is $\boldsymbol{\beta}_k = \boldsymbol{\beta}$ for all k, simplifying the model to:

$$\text{logit}(\theta_{hik}) = \alpha_k + \boldsymbol{x}'_{hi}\boldsymbol{\beta}$$

- Model can also be stated as:

$$\theta_{hik} = \frac{\exp(\alpha_k + \boldsymbol{x}'_{hi}\boldsymbol{\beta})}{1+\exp(\alpha_k + \boldsymbol{x}'_{hi}\boldsymbol{\beta})}$$

$$= \frac{\exp\left\{\alpha_k + \sum_{g=1}^{t} \beta_g x_{hig}\right\}}{1+\exp\left\{\alpha_k + \sum_{g=1}^{t} \beta_g x_{hig}\right\}}$$

where $g = (1, 2, \ldots, t)$ references the explanatory variables

- Main effects model is written in matrix notation as:

$$\begin{bmatrix} \text{logit}(\theta_{111}) \\ \text{logit}(\theta_{112}) \\ \text{logit}(\theta_{121}) \\ \text{logit}(\theta_{122}) \\ \text{logit}(\theta_{211}) \\ \text{logit}(\theta_{212}) \\ \text{logit}(\theta_{221}) \\ \text{logit}(\theta_{222}) \end{bmatrix} = \begin{bmatrix} \alpha_1 & & +\beta_1 & +\beta_2 \\ & \alpha_2 & +\beta_1 & +\beta_2 \\ \alpha_1 & & +\beta_1 & \\ & \alpha_2 & +\beta_1 & \\ \alpha_1 & & & +\beta_2 \\ & \alpha_2 & & +\beta_2 \\ \alpha_1 & & & \\ & \alpha_2 & & \end{bmatrix} = \begin{bmatrix} 1 & 0 & 1 & 1 \\ 0 & 1 & 1 & 1 \\ 1 & 0 & 1 & 0 \\ 0 & 1 & 1 & 0 \\ 1 & 0 & 0 & 1 \\ 0 & 1 & 0 & 1 \\ 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \end{bmatrix} \begin{bmatrix} \alpha_1 \\ \alpha_2 \\ \beta_1 \\ \beta_2 \end{bmatrix}$$

where α_1 = intercept for the 1st cumulative logit

α_2 = intercept for the 2nd cumulative logit

β_1 = incremental effect for females

β_2 = incremental effect for active

- Formulas for Cell Probabilities

		Improvement	
Sex	Treatment	Marked	None
Female	Active	$e^{\alpha_1+\beta_1+\beta_2}/(1+e^{\alpha_1+\beta_1+\beta_2})$	$1/(1+e^{\alpha_2+\beta_1+\beta_2})$
Female	Placebo	$e^{\alpha_1+\beta_1}/(1+e^{\alpha_1+\beta_1})$	$1/(1+e^{\alpha_2+\beta_1})$
Male	Active	$e^{\alpha_1+\beta_2}/(1+e^{\alpha_1+\beta_2})$	$1/(1+e^{\alpha_2+\beta_2})$
Male	Placebo	$e^{\alpha_1}/(1+e^{\alpha_1})$	$1/(1+e^{\alpha_2})$

- Formulas for Model Odds

		Improvement	
Sex	Treatment	Marked vs Some or None	Marked or Some vs None
Female	Active	$e^{\alpha_1+\beta_1+\beta_2}$	$e^{\alpha_2+\beta_1+\beta_2}$
Female	Placebo	$e^{\alpha_1+\beta_1}$	$e^{\alpha_2+\beta_1}$
Male	Active	$e^{\alpha_1+\beta_2}$	$e^{\alpha_2+\beta_2}$
Male	Placebo	e^{α_1}	e^{α_2}

- Odds of marked improvement vs some or none for females compared to males is:

$$\frac{e^{\alpha_1+\beta_1}}{e^{\alpha_1}} = \frac{e^{\alpha_1+\beta_1+\beta_2}}{e^{\alpha_1+\beta_2}} = e^{\beta_1} = \frac{e^{\alpha_2+\beta_1+\beta_2}}{e^{\alpha_2+\beta_2}} = \frac{e^{\alpha_2+\beta_1}}{e^{\alpha_2}}$$

9.2.2 Fitting the Proportional Odds Model with PROC LOGISTIC

```
proc logistic order=data;
  freq count;
  class treatment sex / param=reference;
  model improve = sex treatment / scale=none aggregate;
run;
```

- Ordering of Response Variable:
  ORDER=DATA specified so that values for response are ordered in sequence found in data set (marked, some, and none)

Very important to ensure that ordering is correct when using ordinal data strategies

Output 9.1 Response Profiles

Response Profile

Ordered Value	improve	Total Frequency
1	marked	28
2	some	14
3	none	42

- <u>Test for Proportional Odds:</u>

LOGISTIC performs a score test for appropriateness of proportional odds assumption. Model considered is $\text{logit}\left(\theta_{hik}\right) = \alpha_k + \boldsymbol{x}'_{hi}\boldsymbol{\beta}_k$

Hypothesis tested is that there is a common parameter vector $\boldsymbol{\beta}$ instead of distinct $\boldsymbol{\beta}_k$:

$$H_0 : \boldsymbol{\beta}_k = \boldsymbol{\beta} \text{ for all } k$$

Test has $t \times (r - 2)$ degrees of freedom, where t is the number of explanatory variables and r is the number of response levels

Sample size requirement: need ≈ 5 observations at each outcome at each level of each main effect

Output 9.3 Proportional Odds Test

```
Score Test for the Proportional Odds Assumption

      Chi-Square        DF      Pr > ChiSq

          1.8833         2          0.3900
```

- <u>Test for Goodness of Fit:</u> Similar to evaluation of goodness of fit for dichotomous response logistic regression model. With 80% of observed cell counts ≥ 5, can use counterparts of Q_P and Q_L.

- The model-predicted cell frequencies are:

$$m_{i1} = n_{i+}\left[1+\exp\left(-\left\{\hat{\alpha}_1 + \boldsymbol{x}_i'\hat{\boldsymbol{\beta}}\right\}\right)\right]^{-1}$$

$$m_{i,j+1} = n_{i+}\left\{\left[1+\exp\left(\hat{\alpha}_j + \boldsymbol{x}_i'\hat{\boldsymbol{\beta}}\right)\right]^{-1} - \left[1+\exp\left(\hat{\alpha}_{j+1} + \boldsymbol{x}_i'\hat{\boldsymbol{\beta}}\right)\right]^{-1}\right\}$$

for $j = 1, \ldots, (r-2)$

$$m_{ir} = n_{i+}\left[1+\exp\left(\hat{\alpha}_{r-1} + \boldsymbol{x}_i'\hat{\boldsymbol{\beta}}\right)\right]^{-1}$$

Q_P is distributed as χ^2 with df = $\{(r\text{-}1)(s\text{-}1) - t\}$, where t is the number of explanatory variables.

Output 9.4 Goodness of Fit Statistics

Deviance and Pearson Goodness-of-Fit Statistics

Criterion	DF	Value	Value/DF	Pr > ChiSq
Deviance	4	2.7121	0.6780	0.6071
Pearson	4	1.9099	0.4775	0.7523

Number of unique profiles: 4

Output 9.5 Global Tests

Testing Global Null Hypothesis: BETA=0

Test	Chi-Square	DF	Pr > ChiSq
Likelihood Ratio	19.8865	2	<.0001
Score	17.8677	2	0.0001
Wald	16.7745	2	0.0002

- Can also perform score test for set of additional terms not in the model (e.g., interaction):

```
proc logistic order=data;
   freq count;
   class sex treatment / param=reference;
   model improve = sex  treatment sex*treatment /
       selection=forward start=2;
run;
```

Output 9.6 Score Statistics to Evaluate Goodness of Fit

Residual Chi-Square Test

Chi-Square	DF	Pr > ChiSq
0.2801	1	0.5967

Output 9.7 Type III Analysis of Effects

Type III Analysis of Effects

Effect	DF	Wald Chi-Square	Pr > ChiSq
sex	1	6.2096	0.0127
treatment	1	14.4493	0.0001

Output 9.8 Parameter Estimates

```
                    Analysis of Maximum Likelihood Estimates

                                               Standard
Parameter                DF     Estimate       Error    Chi-Square    Pr > ChiSq

Intercept     marked      1      -2.6671      0.5997       19.7800        <.0001
Intercept     some        1      -1.8127      0.5566       10.6064        0.0011
sex           female      1       1.3187      0.5292        6.2096        0.0127
treatment     active      1       1.7973      0.4728       14.4493        0.0001
```

•<u>Interpretation of Model Parameters:</u>

Parameter	Estimate (SE)	Interpretation
α_1	–2.667 (0.600)	Log odds of marked improvement vs some or none for males with placebo
α_2	–1.813 (0.557)	Log odds of marked or some improvement vs none for males with placebo
β_1	1.319 (0.529)	Increment for both types of log odds due to female sex
β_2	1.797 (0.473)	Increment for both types of log odds due to active drug

Output 9.9 Odds Ratio Estimates

Odds Ratio Estimates

Effect		Point Estimate	95% Wald Confidence Limits	
sex	female vs male	3.739	1.325	10.547
treatment	active vs placebo	6.033	2.388	15.241

The proportional odds model can also include continuous explanatory variables

A plot of the predicted cumulative probabilities for each combination of treatment and sex can be created with the following SAS code:

```
proc logistic order=data plots=effect(polybar
                                    x=treatment*sex);
       freq count;
       class sex treatment /param=reference;
       model improve = sex treatment sex*treatment /
                    selection=forward start=2;
run;
```

Output 9.10


Predicted Cumulative Probabilities for improve
1.00
0.75
0.50
0.25
0.00
Probability
female active
female placebo
male active
male placebo
sex * treatment
improve
marked
some
none

The next PROC LOGISTIC invocation includes the continuous variable AGE and includes the variables of the expanded model in order to obtain the Residual Score statistic.

```
proc logistic descending;
   class sex treatment / param=ref;
   model improve = sex treatment age age*sex
   age*treatment sex*treatment age*age /
   selection=forward slentry=1 start=3 details;
run;
```

The resulting residual chi-square has the value 3.3669 with 4 df (p = 0.4984).

The parameter estimates for the model containing sex, treatment and age are contained in the following table.

Parameter	Estimate (SE)	Interpretation
α_1	–4.683 (1.187)	log odds of marked improvement vs some or none for males with placebo
α_2	–3.784 (1.144)	log odds of marked or some improvement vs none for males with placebo
β_1	1.252 (0.546)	increment for both types of log odds due to female sex
β_2	1.745 (0.477)	increment for both types of log odds due to active drug
β_3	0.038 (0.018)	increment for both types of log odds per year of age

9.2.3 Multiple Qualitative Explanatory Variables

Air Pollution	Job Exposure	Smoking Status	Response I	II	III	IV
Low	No	Non	158	9	5	0
Low	No	Ex	167	19	5	3
Low	No	Current	307	102	83	68
Low	Yes	Non	26	5	5	1
Low	Yes	Ex	38	12	4	4
Low	Yes	Current	94	48	46	60
High	No	Non	94	7	5	1
High	No	Ex	67	8	4	3
High	No	Current	184	65	33	36
High	Yes	Non	32	3	6	1
High	Yes	Ex	39	11	4	2
High	Yes	Current	77	48	39	51

- Level I: no symptoms
- Level II: cough or phlegm less than 3 months/year
- Level III: cough or phlegm more than 3 months/year
- Level IV: cough & phlegm plus shortness of breath more than 3 months/year

- Form three cumulative logits for more severe to less severe responses:

$$\text{logit}\left(\theta_{i1}\right) = \log\left[\frac{\pi_{i4}}{\pi_{i1}+\pi_{i2}+\pi_{i3}}\right]$$

$$\text{logit}\left(\theta_{i2}\right) = \log\left[\frac{\pi_{i4}+\pi_{i3}}{\pi_{i1}+\pi_{i2}}\right]$$

$$\text{logit}\left(\theta_{i3}\right) = \log\left[\frac{\pi_{i4}+\pi_{i3}+\pi_{i2}}{\pi_{i1}}\right],$$

where i = 1,2, …, 12 references the 12 populations

- Main effects proportional odds model:

```
proc logistic descending;
  freq count;
  class air exposure(ref='no') smoking / param=ref;
  model level = air exposure smoking air*exposure
                 air*smoking exposure*smoking /
              selection=forward include=3 scale=none
              aggregate=(air exposure smoking);
run;
```

Output 9.11 Response Profile

Response Profile

Ordered Value	level	Total Frequency
1	4	230
2	3	239
3	2	337
4	1	1283

Output 9.12 Test for Proportional Odds Assumption

Score Test for the Proportional Odds Assumption		
Chi-Square	DF	Pr > ChiSq
12.0745	8	0.1479

Output 9.12 Assessment of Fit

Residual Chi-Square Test		
Chi-Square	DF	Pr > ChiSq
2.7220	5	0.7428

Output 9.14 Goodness of Fit Statistics

Deviance and Pearson Goodness-of-Fit Statistics

Criterion	DF	Value	Value/DF	Pr > ChiSq
Deviance	29	29.9969	1.0344	0.4142
Pearson	29	28.0796	0.9683	0.5137

Number of unique profiles: 12

Output 9.15 Type III Analysis of Effects

Type III Analysis of Effects

Effect	DF	Wald Chi-Square	Pr > ChiSq
air	1	0.1758	0.6750
exposure	1	82.0603	<.0001
smoking	2	209.8507	<.0001

Output 9.16 Parameter Estimates

Analysis of Maximum Likelihood Estimates

Parameter		DF	Estimate	Standard Error	Chi-Square	Pr > ChiSq
Intercept	4	1	-3.8938	0.1779	479.2836	<.0001
Intercept	3	1	-2.9696	0.1693	307.7931	<.0001
Intercept	2	1	-2.0884	0.1633	163.5861	<.0001
air	high	1	-0.0393	0.0937	0.1758	0.6750
exposure	yes	1	0.8648	0.0955	82.0603	<.0001
smoking	cur	1	1.8527	0.1650	126.0383	<.0001
smoking	ex	1	0.4000	0.2019	3.9267	0.0475

Output 9.17 Odds Ratios

Odds Ratio Estimates

Effect		Point Estimate	95% Wald Confidence Limits	
air	high vs low	0.961	0.800	1.155
exposure	yes vs no	2.374	1.969	2.863
smoking	cur vs non	6.377	4.615	8.812
smoking	ex vs non	1.492	1.004	2.216

9.2.4 Partial Proportional Odds Model

The proportional odds assumption may not hold for every explanatory variable in the model. That is, $\beta_k = \beta$ only for some k. In this case, the partial proportional odds model may be applicable.

Example: Cycling Glove Data

		Relief from Symptoms			
Glove Type	**Gender**	**Major**	**Moderate**	**None**	**Total**
Test	Female	12	8	5	**25**
Test	Male	8	14	15	**37**
Gel	Female	5	5	9	**19**
Gel	Male	8	4	20	**32**

A manufacturer wanted to test whether a Test glove performs better than the standard gel glove for relieving numbness and pain after a weeks' worth of bicycle rides. Gender was included as an explanatory variable. SAS code for entering the data is provided on page 275.

The usual proportional odds model was fit with glove and gender as explanatory variables as well as their interaction. The interaction was not significant and was removed. The main effects model was fit using the following code:

```
proc logistic order=data;
       freq count;
       class glove gender / param=ref order=data;
       model relief = glove gender
                     /scale=none aggregate;
run;
```

The test of the proportional odds assumption had p=0.1241, which, although not statistically significant at the α=0.05 level, it can be considered a marginal result.

The partial proportional odds model can be fit which allows for unequal slopes on the explanatory variables while accounting for the ordinal response variable.

```
proc logistic order=data;
        freq count;
        class glove gender / param=ref order=data;
        model relief = glove gender / link=clogit scale=none
                                 aggregate unequalslopes;
        pogender: test genderfemale_major=genderfemale_moderate;
        poglove: test glovetest_major=glovetest_moderate;
run;
```

The specification of LINK=CLOGIT and UNEQUALSLOPES tells SAS that a partial proportional odds model should be fit.

Output 9.22 Parameter Estimates

Analysis of Maximum Likelihood Estimates

Parameter		relief	DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept		major	1	-1.3704	0.3799	13.0128	0.0003
Intercept		moderate	1	-0.6480	0.3362	3.7155	0.0539
glove	test	major	1	0.3209	0.4261	0.5672	0.4514
glove	test	moderate	1	1.0855	0.4022	7.2838	0.0070
gender	female	major	1	0.7189	0.4209	2.9177	0.0876
gender	female	moderate	1	0.9067	0.4143	4.7905	0.0286

The TEST statements are individual tests of the proportional odds assumption for each explanatory variable (i.e. test of equality of the parameters for each type of cumulative logit for an explanatory variable).

Output 9.23 Proportional Odds Test

Linear Hypotheses Testing Results

Label	Wald Chi-Square	DF	Pr>ChiSq
Pogender	0.2311	1	0.6307
poglove	3.9326	1	0.0474

The p-value of 0.0474 for glove suggests proportional odds may not hold for this explanatory variable.

The model can be re-fit to accommodate unequal slopes only for the glove variable using UNEQUALSLOPES=glove:

```
proc logistic order=data;
       freq count;
       class glove gender /param=reference order=data;
       model relief= glove gender /scale=none aggregate
                                 unequalslopes=glove;
run;
```

The Type 3 analysis (not shown) has a 2-df test for the glove effect (since glove has one parameter for each logit type) with p=0.0187.

The test of the gender effect has 1 df (since it has one parameter which represents both logits) and p=0.0263.

Output 9.26 Parameter Estimates

Analysis of Maximum Likelihood Estimates

Parameter		relief	DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept		major	1	-1.4269	0.3669	15.1215	0.0001
Intercept		moderate	1	-0.6123	0.3284	3.4773	0.0622
glove	test	major	1	0.3278	0.4269	0.5895	0.4426
glove	test	moderate	1	1.0672	0.3997	7.1292	0.0076
gender	female		1	0.8180	0.3680	4.9395	0.0263

Output 9.27 Odds Ratios

Odds Ratio Estimates

Effect	relief	Point Estimate	95% Wald Confidence Limits	
glove test vs gel	major	1.388	0.601	3.205
glove test vs gel	moderate	2.907	1.328	6.364
gender female vs male		2.266	1.101	4.661

9.3 Nominal Response: Generalized Logits Model

9.3.1 Methodology

		Learning Style		
School	Program	Self	Team	Class
1	Regular	10	17	26
1	After	5	12	50
2	Regular	21	17	26
2	After	16	12	36
3	Regular	15	15	16
3	After	12	12	20

- Proportional odds model not appropriate since levels of response variable (self, team, class) have no inherent ordering

- Generalized logit defined as:

$$\text{logit}_{hij} = \log\left[\frac{\pi_{hij}}{\pi_{hir}}\right]$$

for $j = 1, 2, \ldots, (r-1)$

- Therefore, generalized logits for above table are:

$$\text{logit}_{hi1} = \log\left[\frac{\pi_{hi1}}{\pi_{hi3}}\right],\ \text{logit}_{hi2} = \log\left[\frac{\pi_{hi2}}{\pi_{hi3}}\right],$$

for $h = 1, 2, 3$ schools, $i = 1, 2$ for programs

- Model fit for generalized logits:

$$\text{logit}_{hij} = \alpha_j + \mathbf{x}'_{hi}\boldsymbol{\beta}_j$$

where j indexes the two logits

9.3.2 Fitting Models to Generalized Logits with PROC LOGISTIC

		Learning Style Preference		
School	Program	Self	Team	Class
1	Regular	10	17	26
1	After	5	12	50
2	Regular	21	17	26
2	After	16	12	36
3	Regular	15	15	16
3	After	12	12	20

A study seeks to assess whether learning style is associated with school or program. Learning style is a nominal variable; a generalized logits model may be fit.

Model with Interaction (see page 282 for data input):

```
proc logistic order=data;
        freq count;
        class school (ref=first) program (ref=last) /param=ref;
        model style = program school school*program /link=glogit
                                scale=none aggregate;
run;
```

The Type 3 table (not shown) has p=0.7827 for the school*program interaction. This term is removed, and the main effects model is fit:

```
proc logistic order=data;
        freq count;
        class school (ref=first) program (ref=last) /param=ref;
        model style = program school / link=glogit
                                scale=none aggregate;
       oddsratio school;
        oddsratio program;
run;
```

Output 9.32 Goodness of Fit

Deviance and Pearson Goodness-of-Fit Statistics				
Criterion	Value	DF	Value/DF	Pr>ChiSq
Deviance	1.7776	4	0.4444	0.7766
Pearson	1.7589	4	0.4397	0.7800

Output 9.33 Analysis of Effects Table

Type 3 Analysis of Effects			
Effect	DF	Wald Chi-Square	Pr>ChiSq
program	2	10.9160	0.0043
school	4	14.8424	0.0050

Table 9.8 Parameter Interpretations:

Model Parameter	Interpretation
α_1	Intercept for $logit_{hi1}$ for School 1, Regular
α_2	Intercept for $logit_{hi2}$ for School 1, Regular
β_1	Incremental Effect for After for $logit_{hi1}$
β_2	Incremental Effect for After for $logit_{hi2}$
β_3	Incremental Effect for School 2 for $logit_{hi1}$
β_4	Incremental Effect for School 2 for $logit_{hi2}$
β_5	Incremental Effect for School 3 for $logit_{hi1}$
β_6	Incremental Effect for School 3 for $logit_{hi2}$

Table 9.9 Parameter Estimates

Variable	Logit (self/class) Coefficient	Logit (self/class) Standard Error	Logit (team/class) Coefficient	Logit (team/class) Standard Error
Intercept	$-1.223(\hat{\alpha}_1)$	0.315	$-0.566(\hat{\alpha}_2)$	0.259
Program	$-0.747(\hat{\beta}_1)$	0.282	$-0.743(\hat{\beta}_2)$	0.271
School 2	$1.083(\hat{\beta}_3)$	0.354	$-0.180(\hat{\beta}_4)$	0.317
School 3	$1.315(\hat{\beta}_5)$	0.384	$0.656(\hat{\beta}_6)$	0.340

Table 9.10 Model-Predicted Odds:

School	Program	Odds: Self/Class	Odds: Team/Class	Odds: Self/Team
1	Regular	e^{α_1}	e^{α_2}	$e^{\alpha_1-\alpha_2}$
1	After	$e^{\alpha_1+\beta_1}$	$e^{\alpha_2+\beta_2}$	$e^{\alpha_1-\alpha_2+\beta_1-\beta_2}$
2	Regular	$e^{\alpha_1+\beta_3}$	$e^{\alpha_2+\beta_4}$	$e^{\alpha_1-\alpha_2+\beta_3-\beta_4}$
2	After	$e^{\alpha_1+\beta_1+\beta_3}$	$e^{\alpha_2+\beta_2+\beta_4}$	$e^{\alpha_1-\alpha_2+\beta_1+\beta_3-\beta_2-\beta_4}$
3	Regular	$e^{\alpha_1+\beta_5}$	$e^{\alpha_2+\beta_6}$	$e^{\alpha_1-\alpha_2+\beta_5-\beta_6}$
3	After	$e^{\alpha_1+\beta_1+\beta_5}$	$e^{\alpha_2+\beta_2+\beta_6}$	$e^{\alpha_1-\alpha_2+\beta_1+\beta_5-\beta_2-\beta_6}$

- Odds ratio of self to class for after vs. regular program:

$$\frac{e^{\alpha_1+\beta_1}}{e^{\alpha_1}} = e^{\beta_1}$$

- Odds ratio of School 1 to School 2 for self/class logit:

$$\frac{e^{\alpha_1}}{e^{\alpha_1+\beta_3}} = e^{-\beta_3}$$

Output 9.35 Odds Ratio Output Estimates and Wald Confidence Intervals

Odds Ratio Estimates

Label	Estimate	95% Confidence Limits	
Style self: school 2 vs 1	2.953	1.476	5.909
Style team: school 2 vs 1	1.197	0.643	2.230
Style self: school 3 vs 1	3.724	1.755	7.902
Style team: school 3 vs 1	1.926	0.990	3.747
Style self: school 2 vs 3	0.793	0.413	1.522
Style team: school 2 vs 3	0.622	0.317	1.219
Style self: after vs regular	0.474	0.273	0.823
Style team: after vs regular	0.476	0.280	0.809

9.3.3 Generalized Logit Model with Continuous Explanatory Variable

Researchers in Florida studied the food wild alligators chose to eat as their primary food. Types of food were fish (F), Invertebrate (I), or other (O). Interest is in assessing how alligator size (length in meters) affects food choice. Since type of food is nominal, a generalized logits model may be used. The data are input using code on Page 289. The following SAS code fits the model:

```
ods graphics on;
proc logistic data=alligator plots=effect;
  model choice(order=data) = length length*length /
               selection=forward include=1 link=glogit;
run;
ods graphics off;
```

Output 9.36 Response Profile

Response Profile

Ordered Value	Choice	Total Frequency
1	I	20
2	F	31
3	O	8

Output 9.37 Residual Chi-Square Test

Residual Chi-Square Test

Chi-Square	DF	Pr>ChiSq
0.4260	2	0.8081

Output 9.39 Parameter Estimates

Analysis of Maximum Likelihood Estimates

Parameter	Choice	DF	Estimate	Standard Error	Wald Chi-Square	Pr>ChiSq
Intercept	I	1	5.6974	1.7938	10.0881	0.0015
Intercept	F	1	1.6177	1.3073	1.5314	0.2159
Length	I	1	-2.4654	0.8997	7.5101	0.0061
Length	F	1	-0.1101	0.5171	0.0453	0.8314

Output 9.40 Odds Ratios

Odds Ratio Estimates and Wald Confidence Intervals

Label	Estimate	95% Confidence Limits	
Choice I: Length	0.085	0.015	0.496
Choice F: Length	0.896	0.325	2.468

For invertebrate compared to other, other has (1/0.085) = 11.76 times the odds of being chosen per meter increase in length (p=0.0061). For fish compared to other, the odds ratio is 1.12 (p=0.8314).

Output 9.41 Model-Predicted Probabilities



9.3.4 Exact Methods for Generalized Logits Model

Exact methods are appropriate when the sample size specifications do not hold (i.e. fewer than 80% of cells have at least 5 of each response per category of each main effect).

Example: Physical Therapy Data

		Strengthening Choice		
Gender	Status	Machines	Free Weights	Bands
Males	Adult	2	13	3
Males	Senior	10	9	3
Females	Adult	3	9	1
Females	Senior	8	0	1

Numerous small counts make it unlikely that an analysis based on asymptotics would be appropriate.

Global Fit Statistics From the Main Effects Analysis:

Testing Global Null Hypothesis: BETA=0

Test	Chi-Square	DF	Pr>ChiSq
Likelihood Ratio	18.9061	4	0.0008
Score	16.9631	4	0.0020
Wald	12.8115	4	0.0122

The p-value for the Wald test is 0.0122, more than ten times larger than the p-value for the likelihood ratio statistic (p=0.0008). This is an indication that the sample sizes are too small for asymptotic methods, and exact approaches should be considered.

```
proc logistic data=pt;
        freq count;
        class gender(ref='females') age(ref='senior') /param=ref;
        model therapy(ref='machines') = gender age /link=glogit;
        exact gender age / joint estimate=both;
Run;
```

Exact Conditional Tests

Effect	Test	Statistic	P-value Exact	P-value Mid
Joint	Score	16.6895	0.0013	0.0013
	Probability	7.155E-7	0.0010	0.0010
gender	Score	5.0830	0.0870	0.0835
	Probability	0.00697	0.0988	0.0953
age	Score	14.5093	0.0003	0.0003
	Probability	0.000032	0.0003	0.0003

Overall exact p=0.0870 for gender, Overall exact p=0.0003 for age.

Output 9.49 Exact Odds Ratios

Exact Odds Ratios

Parameter		Therapy	Estimate	95% Confidence Limits		Two-Sided p-Value
gender	males	bands	4.102	0.493	59.553	0.2715
gender	males	freeweights	4.489	0.933	30.209	0.0641
age	adult	bands	5.076	0.598	53.165	0.1673
age	adult	freeweights	12.467	2.713	89.960	0.0002

Age has a strong influence on the freeweights/machines logit (p=0.0002). Gender has a marginal influence on the freeweights/machines logit (p=0.0641).

There appears to be no influence of either age (p=0.1673) or gender (p=0.2715) on the bands/machines logit.